

FSEAR and MSEAR Scoring

The FSEAR CRF (for females) contains 12 questions, while the MSEAR CRF (for males) contains 14 questions. The various questions are combined into Scales and Subscales, which are then transformed into scaled scores of 0 to 100 so that male and female results can be compared.

Scaled scores are calculated as follows:

$$\text{Scaled_score} = 100 * [(\text{actual raw score} - \text{lowest possible raw score}) / \text{possible raw score range}]$$

FSEAR:

- Scales and Subscales:
 - Sexual Relationship Domain: sum of questions 1–7
 - Impute if 1 response is missing
 - Confidence Domain: sum of questions 8–12
 - No imputation; if any responses are missing, the domain score is missing
 - Self-Esteem subscale: sum of questions 8-10
 - No imputation; if any responses are missing, the subscale is missing
 - Overall Relationship subscale: sum of questions 11 and 12
 - No imputation; if any responses are missing, the subscale is missing
 - Total Score: sum of questions 1–12
 - Impute if 1 or 2 responses are missing

MSEAR:

- Scales and Subscales:
 - Sexual Relationship Domain: sum of questions 1–8
 - Impute if 1 response is missing
 - Confidence Domain: sum of questions 9–14
 - Impute if 1 response is missing
 - Self-Esteem subscale: sum of questions 9–12
 - No imputation; if any responses are missing, the subscale is missing
 - Overall Relationship subscale: sum of questions 13 and 14
 - No imputation; if any responses are missing, the subscale is missing
 - Total Score: sum of questions 1–14
 - Impute if 1 or 2 responses are missing

SAS Code – FSEAR:

```
data FSEAR2;
    set FSEAR1;

    array FSEARs(12) FSEAR_q1-FSEAR_q12;

    non_miss_tot = n(of FSEAR_q1-FSEAR_q12); /* impute if 10 or more non-missing */
    non_miss_sexrel = n(of FSEAR_q1-FSEAR_q7); /* impute if 6 or more non-missing */
    non_miss_confi = n(of FSEAR_q8-FSEAR_q12); /* No imputation - all 5 must be present */

    if non_miss_sexrel ge 6 then
        raw_sear_sexrel = (sum (of FSEAR_q1-FSEAR_q7))*7/non_miss_sexrel;
    sear_sexrel_domain = 100*((raw_sear_sexrel - 7)/28);

    if non_miss_confi ge 5 then
        raw_sear_confi = (sum (of FSEAR_q8-FSEAR_q12))*5/non_miss_confi;
    sear_confi_domain = 100*((raw_sear_confi - 5)/20);

    if non_miss_tot ge 10 then
        raw_sear_tot = (sum (of FSEAR_q1-FSEAR_q12))*12/non_miss_tot;
    sear_total_score = 100*((raw_sear_tot - 12)/48);

    raw_sear_slfeesteem = FSEAR_q8 + FSEAR_q9 + FSEAR_q10;
    sear_slfeesteem_subscale = 100*((raw_sear_slfeesteem - 3)/12);

    raw_sear_ovallrel = FSEAR_q11 + FSEAR_q12;
    sear_ovallrel_subscale = 100*((raw_sear_ovallrel - 2)/8);

    drop non_miss: raw;;
run;

data FSEAR3;
    set FSEAR2;
    array FSEARs(12) FSEAR_q1-FSEAR_q12;

    drop non_miss: raw: ;
    label pid = "Participant ID#"
        vnum = "Visit #"
        FSEAR_q1 = "SEAR - Relaxed about initiating sex"
        FSEAR_q2 = "SEAR - Satisfied with my sexual performance"
        FSEAR_q3 = "SEAR - Sex could be spontaneous"
```

```

FSEAR_q4 = "SEAR - I was likely to initiate sex"
FSEAR_q5 = "SEAR - Confident about performing sexually"
FSEAR_q6 = "SEAR - Satisfied with our sex life"
FSEAR_q7 = "SEAR - Partner unhappy with sexual relations"
FSEAR_q8 = "SEAR - I had good self-esteem"
FSEAR_q9 = "SEAR - Feel that I'm a failure"
FSEAR_q10 = "SEAR - Confident"
FSEAR_q11 = "SEAR - Partner satisfied with relationship"
FSEAR_q12 = "SEAR - I was satisfied with our relationship"
sear_sexrel_domain = "SEAR Sexual Relationship Domain Scaled Score (0-100)"
sear_confid_domain = "SEAR Confidence Domain Scaled Score (0-100)"
sear_slfeesteem_subscale = "SEAR Self-Esteem Subscale Scaled Score (0-100)"
sear_ovallrel_subscale = "SEAR Overall Relationship Subscale Scaled Score (0-100)"
sear_total_score = "SEAR Total Scaled Score (0-100)"
date_fsear = "Date SEAR Survey Completed";

format fsear_q1-fsear_q6 fsear_q8 fsear_q10-fsear_q12 searf.;
format fsear_q7 fsear_q9 sear_revf.;

run;

```

```

data FSEAR4;
set FSEAR3;
rename date_fsear = date_sear;
rename fsear_q1 = sear_relaxed_init;
rename fsear_q2 = sear_satisfied_perf;
rename fsear_q3 = sear_sex_spontaneous;
rename fsear_q4 = sear_likely_init;
rename fsear_q5 = sear_confident_perf;
rename fsear_q6 = sear_satisfied_sexlife;
rename fsear_q7 = sear_partner_unhappy;
rename fsear_q8 = sear_self_esteem;
rename fsear_q9 = sear_failure;
rename fsear_q10 = sear_confident;
rename fsear_q11 = sear_partner_satisfied;
rename fsear_q12 = sear_satisfied_relat;

run;

```

SAS Code – MSEAR:

```
data MSEAR2;
    set MSEAR1;

    array msears(14) msear_q1-msear_q14;

    non_miss_tot = n(of msear_q1-msear_q14); /* impute if 12 or more non-missing */
    non_miss_sexrel = n(of msear_q1-msear_q8); /* impute if 7 or more non-missing */
    non_miss_confi = n(of msear_q9-msear_q14); /* impute if 5 or more non-missing */

    if non_miss_sexrel ge 7 then
        raw_sear_sexrel = (sum (of msear_q1-msear_q8))*8/non_miss_sexrel;
    sear_sexrel_domain = 100*((raw_sear_sexrel - 8)/32);

    if non_miss_confi ge 5 then
        raw_sear_confi = (sum (of msear_q9-msear_q14))*6/non_miss_confi;
    sear_confid_domain = 100*((raw_sear_confi - 6)/24);

    if non_miss_tot ge 12 then
        raw_sear_tot = (sum (of msear_q1-msear_q14))*14/non_miss_tot;
    sear_total_score = 100*((raw_sear_tot - 14)/56);

    raw_sear_slfesteem = msear_q9 + msear_q10 + msear_q11 + msear_q12;
    sear_slfesteem_subscale = 100*((raw_sear_slfesteem - 4)/16);

    raw_sear_ovallrel = msear_q13 + msear_q14;
    sear_ovallrel_subscale = 100*((raw_sear_ovallrel - 2)/8);

    drop non_miss: raw;;

run;

data MSEAR3;
    set MSEAR2;
    array msears(14) msear_q1-msear_q14;

    drop non_miss: raw: ;
    label pid = "Participant ID#"
        vnum = "Visit #"
        MSEAR_q1 = "SEAR - Relaxed about initiating sex"
        MSEAR_q2 = "SEAR - Confident erection would last"
        MSEAR_q3 = "SEAR - Satisfied with my sexual performance"
```

```

MSEAR_q4 = "SEAR - Sex could be spontaneous"
MSEAR_q5 = "SEAR - I was likely to initiate sex"
MSEAR_q6 = "SEAR - Confident about performing sexually"
MSEAR_q7 = "SEAR - Satisfied with our sex life"
MSEAR_q8 = "SEAR - Partner unhappy with sexual relations"
MSEAR_q9 = "SEAR - I had good self-esteem"
MSEAR_q10 = "SEAR - Felt like a whole man"
MSEAR_q11 = "SEAR - Feel that I'm a failure"
MSEAR_q12 = "SEAR - Confident"
MSEAR_q13 = "SEAR - Partner satisfied with relationship"
MSEAR_q14 = "SEAR - I was satisfied with our relationship"
sear_sexrel_domain = "SEAR Sexual Relationship Domain Scaled Score (0-100)"
sear_confid_domain = "SEAR Confidence Domain Scaled Score (0-100)"
sear_slfeesteem_subscale = "SEAR Self-Esteem Subscale Scaled Score (0-100)"
sear_ovallrel_subscale = "SEAR Overall Relationship Subscale Scaled Score (0-100)"
sear_total_score = "SEAR Total Scaled Score (0-100)"
date_MSEAR = "Date SEAR Survey Completed";

format msear_q1-msear_q7 msear_q9 msear_q10 msear_q12-msear_q14 searf.;
format msear_q8 msear_q11 sear_revf.;

run;

data MSEAR4;
set MSEAR3;
rename date_msear = date_sear;
rename msear_q1 = sear_relaxed_init;
rename msear_q2 = sear_erection_last;
rename msear_q3 = sear_satisfied_perf;
rename msear_q4 = sear_sex_spontaneous;
rename msear_q5 = sear_likely_init;
rename msear_q6 = sear_confident_perf;
rename msear_q7 = sear_satisfied_sexlife;
rename msear_q8 = sear_partner_unhappy;
rename msear_q9 = sear_self_esteem;
rename msear_q10 = sear_whole_man;
rename msear_q11 = sear_failure;
rename msear_q12 = sear_confident;
rename msear_q13 = sear_partner_satisfied;
rename msear_q14 = sear_satisfied_relat;
run;

```